



## アバランシェTVSダイオード のSPICEマクロモデル

### はじめに

SPICEマクロモデルを使用すると、TVSアバランシェ・ダイオードの電流-電圧特性を高精度でシミュレートできます。これらのモデルは、サージ保護回路の性能の分析と最適化に使用できます。TVSマクロモデルは、複数の標準的なSPICEデバイスを組み合わせてサブ回路を形成する方法で作成されています。


### データシートの仕様

TVSマクロモデルを分析するために最初に必要な事項は、データシートに掲載されているデバイス仕様を確認することです。Figure 1に、単方向アバランシェTVSダイオードの電流および電圧の定義を示します。

ON Semiconductor®

[www.onsemi.jp](http://www.onsemi.jp)

## APPLICATION NOTE



$I_F$	Forward current	$V_{BR}$	Breakdown voltage @ $I_T$
$V_F$	Forward voltage @ $I_F$	$I_{PP}$	Maximum reverse peak pulse current (typically specified with either the $8 \times 20 \mu s$ or $10 \times 1000 \mu s$ surge pulse)
$I_R$	Reverse leakage current	$V_C$	Clamping voltage @ $I_{PP}$
$V_{RWM}$	Reverse working voltage @ $I_R$ ( $V_{RWM} (typ.) \cong 0.8 \times V_{BR}$ )		
$I_T$	Test current		

Figure 1. Definition of the Current and Voltage Data Sheet Specifications

データシートの他の重要な仕様として、静電容量とピーク電力定格を上げることができます。ダイオードの静電容量は通常、バイアス電圧が0 Vdc、AC信号が50 mV、1.0 MHzの状態の規定されます。小型パッケージの場合は通常、電力定格を $8 \times 20 \mu\text{s}$  (立ち上がり時間 $\times$ パルス持続時間)で定義するのに対し、大型パッケージの場合はデバイス定義で $10 \times 1000 \mu\text{s}$ のサージ・パルスを使用することが多くあります。ワット単位の電力を測定するには、サージ電流( $I_{pp}$ )波形とクランプ電圧( $V_C$ )波形を互いに乗算します。

マクロモデル・サブ回路

TVSダイオードのマクロモデルは、複数の標準SPICEデバイスを組み合わせてサブ回路を形成する方法で作成されます。Figure 2にマクロモデルの回路図を示します。Appendix Iに、1SMB28AおよびNUP2105マクロモデルのPSPICEネットリストを示します。TVSマクロモデルは、参考資料[3]と[4]に掲載されているツェナ・ダイオード・モデルに基づいています。参考資料[1]と[2]に代替のTVSダイオードSPICEモデルも示します。




Figure 2. TVS Avalanche Diode SPICE Macro-Model

順方向領域

ダイオード $D_1$ は、電圧 $V_D$ が0より大きいときの主要コンポーネントです。TVSダイオードの順方向バイアス特性は、 $D_1$ の飽和電流( $I_S$ )、放出係数( $N$ )、および直列抵抗( $R_S$ )によって決まります。順方向バイアス領域の電流式は次のように表現できます。

$$I_D = I_F + I_L + I_R$$

$$= I_{F,D1} + \frac{V_D}{R_L} + I_{S,D2}$$

$$I_L \ \& \ I_R \ll I_F$$

$$\therefore I_D \cong I_{F,D1} \cong I_{S,D1} \left[ e^{\left( \frac{V_{D1}}{\eta V_T} \right)} - 1 \right] \cong I_{S,D1} \left[ e^{\left( \frac{V_{D1}}{\eta V_T} \right)} \right]$$

ここで：

$$V_T = \frac{kT}{q} \cong 26 \text{ mV} @ 25^\circ\text{C}$$

$k$  = Boltzmann's constant  
 $= 1.38 \times 10^{-23} \text{ joules/5K}$

$q$  = Electronic charge  
 $= 1.6 \times 10^{-19} \text{ coulombs}$

$T$  = Absolute temperature (Kelvin)

漏れ領域

漏れ領域つまり逆バイアス領域は、電圧 $V_D$ が0 V ~ ブレークダウン電圧( $V_{BR}$ )の間にある範囲と定義されます。電流 $I_F$ と $I_R$ は、 $I_L$ に比べて小さな値になります。これはダイオード $D_1$ と $D_2$ が逆バイアスされるためです。したがって、リーク電流は $V_D/R_L$ で近似できます。

$$I_D = I_F + I_L + I_R$$

$$= I_{S,D1} + \frac{V_D}{R_L} + I_{S,D2}$$

$$I_F \ \& \ I_R \ll I_L$$

$$\therefore I_D \cong \frac{V_D}{R_L}$$

ブレークダウン領域

ブレークダウン領域は、 $EV_1$ 、 $D_2$ 、および $R_Z$ でモデル化されます。電圧が $EV_1$ に $D_2$ の順方向電圧を加えた値を上回ると、この経路を電流が流れます。ブレークダウン電圧 $V_{BR}$ は、テスト電流 $I_T$ で規定され、 $I_{BV}$ と $R_{BV}$ の積に等しくなります。 $D_3$ を使用して、 $D_2$ の電圧降下を補償します。クランプ電圧( $V_C$ )

は、電流 $I_{PP}$ で規定され、次に示すように、 $V_{EV1}$ 、 $R_Z$ 、 $D_2$ の電圧の合計に等しくなります。

$$I_D \cong I_S \left[ e^{\left( \frac{V_D}{\eta V_T} \right)} \right] \quad \therefore V_D \cong \eta V_T \left[ \ln \left( \frac{I_D}{I_S} \right) \right]$$

$$\begin{aligned} V_C @ I_{PP} &= V_{EV1} + V_{D2} + V_{RZ} \\ &= \left[ V_{BR} - \eta_3 V_T \ln \left( \frac{I_T}{I_{S3}} \right) \right] + \eta_2 V_T \ln \left( \frac{I_{PP}}{I_{S2}} \right) \\ &\quad + (I_{PP} R_Z) \end{aligned}$$

$$V_{EV1} = V_{BR} = V_D @ I_T = I_{BV} R_{BV}$$

**インピーダンス特性**

TVSダイオードのインピーダンスは、誘導成分、容量成分、抵抗成分で構成されます。誘導成分をモデル化すると、ICパッケージのインダクタンスに起因するオーバシュート・パルスの振幅( $V = L (\Delta I/\Delta t)$ )が確実にシミュレートされます。容量成分をマッチングさせると、クランプ波形の形状を予測しやすくなります。デバイスの電力能力を予測するうえで、高精度の抵抗成分を式に含めることが重要です。

**ACモデル**

TVSダイオードのインピーダンスは、ネットワーク・アナライザを使用して測定できます。次に、測定したインピーダンスの実数成分と虚数成分を使用して、等価小信号モデルつまりACモデルを求める

ことができます。ACモデルは直列接続された抵抗( $R_S$ )、インダクタ( $L_S$ )、コンデンサ( $C_S$ )で構成されます。 $R_S$ は、複素インピーダンスの実数部分に等しく、共振周波数( $f_R$ )で測定されます。周波数が $f_R$ では、インピーダンスは純粋に抵抗成分のみになります。これは $L_S$ と $C_S$ のインピーダンスは大きさが等しく、極性が正反対であるためです。 $C_S$ は通常、1.0 MHzの周波数で静電容量を測定して求めます。 $L_S$ は共振周波数から求め、最小インピーダンスに相当します。Table 1に、ACモデルのインピーダンス成分をSPICEマクロモデルに統合する方法を示します。ACモデルの設計式は次のとおりです。

$$Z_R = R; \quad Z_C = \frac{-j}{\omega C}; \quad Z_L = \omega L; \quad \omega = 2\pi f$$

$$Z = R_{eqv} + jX_{eqv}$$

$$Z = \sqrt{R_{eqv}^2 + X_{eqv}^2}$$

$$= \sqrt{R_S^2 + \left( 2\pi f L_S - \frac{1}{2\pi f C_S} \right)^2}$$

$$@ f_R |Z_L| = |Z_C|$$

$$= R_S \Rightarrow @ f_R; \quad Z = Z_{Min} = R_S$$

$$C_S \Rightarrow @ 1 \text{ MHz } Z_{CS} \gg Z_{subLS} \quad \therefore C_S \cong \frac{1}{2\pi f Z}$$

$$L_S \Rightarrow f_R = \frac{1}{2\pi \sqrt{L_S C_S}} \quad \therefore L_S = \frac{1}{4\pi^2 f_R^2 C_S}$$

**Table 1. CORRELATION OF THE AC AND MACRO-MODEL COMPONENTS**

AC Model Component	Equivalent Macro-Model Component	Comments
$R_S$	$R_Z + D_{2\_RS}$	<ul style="list-style-type: none"> <li>Typically <math>D_{2\_RS} = 0</math>; thus, <math>R_S = R_Z</math></li> <li><math>R_Z \propto</math> clamping voltage <math>V_C</math></li> <li><math>R_Z \propto</math> 1/power rating</li> </ul>
$L_S$	$L$	<ul style="list-style-type: none"> <li><math>L</math> produces a short overshoot pulse due to <math>V = L (\Delta I/\Delta t)</math></li> </ul>
$C_S$	$D_{1\_CJ0}$	<ul style="list-style-type: none"> <li><math>D_{1\_CJ0}</math> is specified at a 0 V and decreases as the reverse bias voltage increases</li> </ul>

**測定テストとACモデルによるインピーダンス・データ**

Figure 3とFigure 4に、1SMB28AとNUP2105のインピーダンスを示します。TVSダイオードのインピーダンスは、Figure 3に示すように、バイアス電圧の関数です。また、DCバイアス電圧が大きくなると静電容量が減少し、共振周波数( $f_R$ )は高くなります。TVS

ダイオードは、周波数が比較的低い場合は1個のコンデンサとしてモデル化できますが、周波数が共振周波数に近付くときは、ICパッケージのインダクタンスを式に含める必要が生じます。Table 2に、1SMB28AとNUP2105の実際に測定したインピーダンスと、ACモデル・パラメータの要約を示します。

## AND8254/D




Figure 3. Impedance Characteristic of the 1SMB28A Unidirectional TVS Diode




Figure 4. Impedance Characteristic of the NUP2105 Bidirectional TVS Diode

### 抵抗

インピーダンスの実数部分つまり抵抗成分は、ACモデルでは $R_S$ 、SPICEモデルでは $R_Z$ としてモデル化されます。抵抗はデバイスの電力定格を決定する主要な要因であり、ICパッケージ・リードをシリコン・ダイに接続するのに使用する手法の機能です。SMBリードの比較的大きなパッド・サイズでは、リードからシリコンへの接続部に接触面積を大きく確保できるので抵抗は小さくなります。さらに、SMBの大きなリード・サイズにより熱抵抗が小さくなり、リードからPCBのマウント・パッドに達するリードを通じて消費できる熱エネルギー量も増加します。SMDデバイスと比較すると、SOT-23のリードからシリコンへの接続は比較的抵抗が大きくなっています。

サージ・パルスの高エネルギーが原因で、TVSダイオードの接合部温度は周囲温度に比べて1桁大きい値で上昇するおそれがあります。TVSダイオードは高い接合部温度に耐えるように設計されていますが、ブレイクダウン電圧( $V_{BR}$ )と抵抗は、それぞれの公称値より大きくなります。高いダイ温度をシミュレートする1つのオプションは、マクロモデルの $R_Z$ 値を大きくし、シミュレートされたクランプ電圧が、 $8 \times 20 \mu s$ 、または $10 \times 1000 \mu s$ のサージ・テストなど、特定パルスにおけるベンチ・テスト値に一致するようにすることです。 $R_Z$ を大きくすると、Figure 5に示すように、シミュレートされる最小インピーダンス( $Z_{Min}$ )は大きくなりますが、共振周波数は変化しません。




Figure 5. The Increase in the 1SMA28A’s Junction Temperature Produced by a High Energy Surge Pulse can be Modeled by Increasing the Magnitude of  $R_Z$  from the Nominal Value of 0.1 to 0.65  $\Omega$

静電容量とインダクタンス

静電容量( $C_S$ )とインダクタンス( $L_S$ )は、TVSダイオードのインピーダンスのうち、虚数部分つまりリアクタンス成分を形成します。静電容量はシリコン接合部領域のサイズに比例します。SMBデバイスはSOT-23より大きいダイを収容しています。したがって、通常はSMBデバイスの方がSOT-23デバイスより共振周波数が低くなっています。さらに、双方向ダイオードの静電容量は、同等な単方向デバイスの

静電容量の半分になります。双方向ダイオードは単方向ダイオードを2個直列接続して製作されています。したがって、単方向ダイオードより静電容量が小さくなります。インダクタンス成分は、パッケージのリードとシリコン・ダイの間にあるボンディング接続によって形成されます。 $L_S$ の大きさは、1SMB28AとNUP2105の各TVSダイオードで互いに類似しています。

Table 2. THE SMALL  $R_S$  AND LARGE  $C_S$  TERMS OF THE 1SMB28A ACCOUNT FOR THE DEVICES HIGH POWER RATING. THE SMALL CAPACITANCE OF THE NUP2105 RESULTS IN A HIGH RESONANT FREQUENCY

Part Number	Package and Schematic	Power Rating	$f_R$ (MHz)	Bias Voltage	AC Model		
					$R_S$ ( $\Omega$ )	$L_S$ (nH)	$C_S$ (pF)
1SMB28A		600 W (10 × 1000 $\mu$ s)	146	0 Vdc	0.12	2.44	486
			276	28 Vdc	0.14	2.44	137
NUP2105		350 W (8 × 20 $\mu$ s)	616	0 Vdc	1.28	2.48	26.4

シミュレーション・テストの結果

10 × 1000  $\mu$ sのサージ・テストに対する1SMB28A TVSダイオードのクランプ性能をFigure 6に示します。SPICEシミュレーションでは、 $R_Z$ 値として、ネットワーク・アナライザで測定した抵抗値0.1  $\Omega$ ではなく、0.65  $\Omega$ を使用しました。抵抗値が大きくなると、高エネルギー・サージに対するクランプ電圧

( $V_C$ )の精度が高くなりますが、比較的小さなエネルギーのパルスに対しては、ベンチ測定より大きい $V_C$ がシミュレートされることとなります。マクロモデルの将来の拡張では、熱モデルを統合して、自己加熱に起因するTVSデバイスの接合部温度の上昇をシミュレートする予定です。




Figure 6. SPICE Predicts a Maximum Clamping Voltage of 42.5 V if  $R_Z$  is equal to 0.65  $\Omega$ . The Bench Test Value is 42.4 V

$8 \times 20 \mu\text{s}$ のサージ・テストに対するNUP2105 TVSダイオードのクランプ性能をFigure 7に示します。マクロモデルでは、 $R_Z$ の値として、ACモデルから決定した値である1.28  $\Omega$ を使用しました。

$10 \times 1000 \mu\text{s}$ パルスと比較して、 $8 \times 20 \mu\text{s}$ サージのほうが持続時間が短いので、シミュレートした $V_C$ は比較的測定値に近くなります。




Figure 7. SPICE Predicts a Maximum Clamping Voltage of 39.2 V. The Bench Test Measured Value is 40.8 V

**SPICEの制限**

マクロモデルを使用すると、ほとんどのアプリケーションでTVSアバランシェ・ダイオードの電流-電圧特性をSPICEで高精度に表現できます。SPICEはサ

ージ抑制回路を解析するための強力な設計ツールですが、シミュレーションをハードウェア開発テストの代わりとして使用しないでください。マクロモデルの制限の要約をTable 3に示します。

Table 3. SIMULATION LIMITS OF TVS DIODE MACRO-MODELS

Region	Key Design Parameter	Limitation
Forward	Forward Voltage ( $V_F$ )	<ul style="list-style-type: none"> <li><math>V_F</math> is typically specified as a maximum value at a single current point in the data sheet</li> <li>The accuracy is enhanced if two typical test points are used</li> </ul>
Leakage	Leakage Current ( $I_L$ )	<ul style="list-style-type: none"> <li><math>I_L</math> is modeled as a linear function of the bias voltage</li> <li>Measured <math>I_L</math> data varies as an exponential function of the bias voltage</li> </ul>
Breakdown	Clamping Voltage ( $V_C$ )	<ul style="list-style-type: none"> <li><math>\Delta V_C</math> due to self heating is not modeled</li> <li>Overcurrent failures are not modeled</li> </ul>

### 参考資料

- [1] Bley, M., Filho, M. and Raizer, A., Modeling Transient Discharge Suppressors”, *IEEE Potentials*, August/September 2004.
- [2] Hageman, S., “Model Transient Voltage Suppression Diodes”, *MicroSim Application Notes*, 1997.
- [3] Lepkowski, J., “AND8250 – Zener Macro-Models Provide Accurate SPICE Simulations”, ON Semiconductor, 2005.
- [4] Wong, S.; Hu, C. and Chan, S., “SPICE Macro-Model for the Simulation of Zener Diode Current-Voltage Characteristics”, *International Journal of Electronics*, Volume 71, No. 24, August, 1991.


APPENDIX I: MACRO-MODEL SPICE NETLISTS

```

*****
*   1SMB28A PSPICE macro-model
*   Uni-directional TVS avalanche diode, SMB package, VBR = 32.75 V
*****
*           Anode  Cathode
.SUBCKT SMB28A  7      1
*****
*   Forward Region
*   D1's CJO term models the capacitance
D1 2 1 MDD1
.MODEL MDD1 D IS = 1.83708e-14 N = 1 XTI = 1 RS = 0.2
+ CJO = 486e-12 TT = 5e-10
*****
*   Leakage Region
*   RL models leakage current (IL)
*   MDR temp. coef. model ΔIL/ΔT
RL 1 2 MDR 5.64e+06
.MODEL MDR RES TC1 = 0 TC2 = 0
*****
*   Reverse Breakdown Region
*   RZ models the ΔI/ΔV slope
*   The small signal impedance is equal to 0.1 Ω
*   A RZ value of 0.65 Ω matches the clamping voltage at max. current
*   Increasing RZ models the self-heating from the energy of a surge event
RZ 2 3 0.65
D2 4 3 MDD2
.MODEL MDD2 D IS = 2.5e-15 N = 0.5
*   Breakdown Voltage (VBR) = IBV × RBV
EV1 1 4 6 8 1
IBV 0 6 0.001
RBV 6 0 MDRBV 32750
*   MDRBV temp. coef. model ΔVBR/ΔT
.MODEL MDRBV RES TC1 = 0.00098
D3 8 0 MDD2
IT 0 8 0.001
*****
*   L models the lead-to-silicon connection package inductance
L 7 2 2.44e-9
*
.ENDS SMB28A
*****

```

**NUP2105 Macro-Model**



NUP2105  
Dual Line Bi-directional TVS Diodes  
SOT-23 Package

```

*****
*   NUP2105 PSPICE macro-model
*   Bi-directional TVS avalanche diode, SOT-23 package, VBR = 26.4 V
*   Model simulates 1 of the 2 I/O lines
*****
*           DA Cathode  DB Cathode  DA,B Common Anode
.SUBCKT NUP2105  1      2      3

```

# AND8254/D

```
* Bidirectional devices are formed from two uni-directional devices
X1 3 1 HALFNU2105
X2 3 2 HALFNU2105
.ENDS NUP2105
*****
* Model HALFNU2105 represents one bi-directional pair of a dual device
* Anode Cathode
.SUBCKT HALFNU2105 7 1
* Forward Region
* D1's CJO term models the capacitance
D1 2 1 MDD1
.MODEL MDD1 D IS = 1.83708e-14 N = 1 XTI = 1 RS = 0.2
+ CJO = 26.4e-12 TT = 1e-08
*****
* Leakage Region
* RL models leakage current (IL)
* MDR temp. coef. model ΔIL/ΔT
RL 1 2 MDR 4.32244e+08
.MODEL MDR RES TC1=0 TC2=0
*****
* Reverse Breakdown Region
* RZ models the ΔI/ΔV slope
RZ 2 3 1.28
D2 4 3 MDD2
.MODEL MDD2 D IS = 2.5e-15 N = 0.5
* Breakdown Voltage (VBR) = IBV × RBV
EV1 1 4 6 8 1
IBV 0 6 0.001
RBV 6 0 MDRBV 26357.1
* MDRBV temp. coef. model ΔVBR/ΔT
.MODEL MDRBV RES TC1 = 0.00096
D3 8 0 MDD2
IT 0 8 0.001
*****
* L models the lead-to-silicon connection package inductance
* L is distributed between two diodes for bi-directional diodes
L 7 2 1.24e-9
*
.ENDS halfnup2105
*****
*****
```

ON Semiconductor及びON SemiconductorのロゴはON Semiconductorという商号を使うSemiconductor Components Industries, LLC 若しくはその子会社の米国及び/または他の国における商標です。ON Semiconductorは特許、商標、著作権、トレードシークレット(営業秘密)と他の知的所有権に対する権利を保有します。ON Semiconductorの製品/特許の適用対象リストについては、以下のリンクからご覧いただけます。[www.onsemi.com/site/pdf/Patent-Marketing.pdf](http://www.onsemi.com/site/pdf/Patent-Marketing.pdf) ON Semiconductorは通告なしで、本書記載の製品の変更を行うことがあります。ON Semiconductorは、いかなる特定の目的での製品の適合性について保証しておらず、また、お客様の製品において回路の応用や使用から生じた責任、特に、直接的、間接的、偶発的な損害など一切の損害に対して、いかなる責任も負うことはできません。お客様は、ON Semiconductorによって提供されたサポートやアプリケーション情報の如何にかかわらず、すべての法令、規制、安全性の要求あるいは標準の遵守を含む、ON Semiconductor製品を使用したお客様の製品とアプリケーションについて一切の責任を負うものとします。ON Semiconductorデータシートや仕様書に示される可能性のある「標準的」パラメータは、アプリケーションによっては異なることもあり、実際の性能も時間の経過により変化する可能性があります。「標準的」パラメータを含むすべての動作パラメータは、ご使用になるアプリケーションに応じて、お客様の専門技術者において十分検証されるようお願い致します。ON Semiconductorは、その特許権やその他の権利の下、いかなるライセンスも許諾しません。ON Semiconductor製品は、生命維持装置や、いかなるFDA(米国食品医薬品局)クラス3の医療機器、FDAが管轄しない地域において同一もしくは類似のものと同様に分類される医療機器、あるいは、人体への移植を対象とした機器における重要部品などへの使用を意図した設計はされておらず、また、これらを使用対象としておりません。お客様が、このような意図されたものではない、許可されていないアプリケーション用にON Semiconductor製品を購入または使用した場合、たとえ、ON Semiconductorがその部品の設計または製造に関して過失があったと主張されたとしても、そのような意図せぬ使用、また未許可の使用に関連した死傷等から、直接、又は間接的に生じるすべてのクレーム、費用、損害、経費、および弁護士料などを、お客様の責任において補償をお願いいたします。また、ON Semiconductorとその役員、従業員、子会社、関連会社、代理店に対して、いかなる損害も与えないものとします。ON Semiconductorは雇用機会均等/差別撤廃雇用主です。この資料は適用されるあらゆる著作権法の対象となっており、いかなる方法によっても再販することはできません。

## PUBLICATION ORDERING INFORMATION

**LITERATURE FULFILLMENT:**  
Literature Distribution Center for ON Semiconductor  
19521 E. 32nd Pkwy, Aurora, Colorado 80011 USA  
**Phone:** 303-675-2175 or 800-344-3860 Toll Free USA/Canada  
**Fax:** 303-675-2176 or 800-344-3867 Toll Free USA/Canada  
**Email:** [orderlit@onsemi.com](mailto:orderlit@onsemi.com)

**N. American Technical Support:** 800-282-9855 Toll Free  
USA/Canada  
**Europe, Middle East and Africa Technical Support:**  
Phone: 421 33 790 2910  
**Japan Customer Focus Center**  
Phone: 81-3-5817-1050

**ON Semiconductor Website:** [www.onsemi.com](http://www.onsemi.com)  
**Order Literature:** <http://www.onsemi.com/orderlit>

For additional information, please contact your local Sales Representative